
Povzetek

Predstavljeni so bistveni prijemi pri programiranju
računalnǐske grafike z grafǐcno knjǐznico OpenGL. Ta
dokument naj bi bil osnova za vaje pri predmetih RPK,
OPK, PK in izdelavo seminarjev s tega področja. Ker
študenti FS pridobijo znanje programiranja v jeziku
Fortran, so primeri podani za ta jezik. To pa ne pomeni,
da je teorija omejena le na ta jezik, saj brez bistvenih
popravkov kode lahko pišemo tudi v jeziku C in C++.

1 Uvod

Za modeliranje posebnih modelov običajno ni mǒzno
uporabiti splǒsno namenskih grafičnih orodij. To
se pokǎze predvsem pri vizualizaciji iňzenirskih
prerǎcunov. Rezultati modeliranja običajno niso le
funkcije ampak kompleksni objekti kot so grafi, vodi,
hierarhǐcne strukture, animacije gibanja, mehanizmi,
kontrola poti, volumski modeli posebnih oblik, ...

Skozi razvoj rǎcunalnikov so se uvajali različni stan-
dardi za grafiko. Od zǎcetnikov kot je GKS in nje-
govega naslednika PHIGS je ostal leše spomin. To
pa predvsem zaradi zahtevnosti implementacije in za-
prtosti kode. Kot edini ,,odprti“ standard obstaja le
OpenGL, ki ga je najprej uvedel SGI na svojih grafično
podprtih delovnih postajah. Poleg OpenGL obstaja
tudi Microsoftov Direct3D, ki pa je omejen na PC
računalnike z Windows in ni tako enostaven za uporabo
kot OpenGL, ki se je zaradi svoje odprtosti in zmoglji-
vosti uveljavil na vseh operacijskih sistemih in strojnih
platformah.

Jezik OpenGL je konstruiran kot strojno-neodvisen
vmesnik med programsko kodo in grafičnim po-
spěsevalnikom. Strojna neodvisnost jezika OpenGL po-
neni tudi to, da v specifikaciji jezika ni podpore za nad-
zor okenskega sistema in dogodkov (events) pri interak-
tivnem programiranju. Za tak nadzor so za vsak opera-
cijski sistem izdelani vmesniki, ki povezujejo OpenGL

stroj z okenskim sistemom.
Zaradi specifǐcnosti razlǐcnih okenskih sistemov

(Windows, Xwindow, MacOS, BeOS) je potrebno za
vsak sistem uporabiti posebne prijeme pri klicanje
OpenGL ukazov. Da bi vseeno lahko posali programe,
s sicer omejeno funkcionalnostjo uporabniškega vme-
snika, se je izdelala knjižnica GLUT (GL UTility), ki
vse razlike med operacijskimi sistemi kompenzira in
vpeljuje skupen nǎcin manipuliranja z dogodki (events).
S knjižnico GLUT je tako mogǒce pisati prenosljive
programe, ki so enostavni za programiranje in dovolj
zmogljivi za nezahtevne uporabniške vmesnike.

2 Enostavni OpenGL program

Osnovni jezik OpenGL je podan v knjižnici GL. Bolj
zahtevni primitivi se gradijo z knjižnico GLU (GL Uti-
lity) v kateri so podprogrami, ki uporabljajo rutine GL.
Rutine GLU vsebujejo věc GL ukazov, ki pa so splǒsno
uporabni in so bili zato standardizirani.

2.1 Dogodki

Vsi okenski vmesniki delujejo na principu dogod-
kov (events). To so signali okenskega sistema, ki se
pǒsiljajo programu. Nǎs program je tako v celoti od-
govoren za vsebino okna. Okenski sistem mu le dodeli
podrǒcje (okno) katero vsebino mora popolnoma nad-
zorovati. Poleg dodeljenega področja pa okenski sistem
pǒsilja še sporǒcila nǎsemu programu. Najbolj pogosta
sporǒcila so:

display Prosim obnovi (nariši) vsebino okna. Věc
možnih primerov je, da se to zgodi. Lahko je drugo
okno odkrilo del nǎsega okna, okno se je prema-
knilo na omizju ali pa se je ponovno prikazalo po
tem ko je bilo ikonizirano. Prestrezanje tega dogo-
dek je obvezno saj mora prav vsak program poskr-
beti, da se vsebina okna obnovi.

1



reshape Velikost/oblika okna se je spremenila.
Porǎcunaj vsebino okna za novo velikost. Ta
dogodek se zgodi, kadar uporabnik z miško
spremeni velikost okna.

keyboard Pritisnjena je bila tipka na tipkovnici.

mouse Stanje gumbov na miški se je spremenilo. Upo-
rabnik je pritisnil ali sprostil enega od gumbov.

motion Uporabnik premika miško ob pritisnjenem
gumbu.

timer Program zahteva sporočilo po preteku po
določenegǎcasa, da bo popravil vsebino okna. Pri-
merno je zǎcasovne simulacije.

Seveda poleg naštetih dogodkov obstajajǒse drugi do-
godki, za katere lahko skrbi naš program. Ni pa po-
trebno da nǎs program skrbi za vse naštete dogodke.
Običajno mora program povedati okenskem sistemu, za
katere dogodke bo skrbel in za te dogodke mu bo sistem
tudi pǒsiljal sporǒcila.

2.2 GLUT
Za abstrakcijo dogodkov okenskega sistema v našem
primeru skrbi knjǐznica GLUT. Primer minimalnega
programa, ki narǐsečrto, je naslednji:

subroutine display
include ’GL/fgl.h’
implicit none
call fglClear(GL_COLOR_BUFFER_BIT)
call fglColor3f(1.0, 0.4, 1.0)
call fglBegin(GL_LINES)
call fglVertex2f(0.1,0.1)
call fglVertex3f(0.8,0.8,1.0)
call fglEnd
call fglFlush
end

program crta
external display
include ’GL/fglut.h’
call fglutInit
call fglutInitDisplayMode(GLUT_SINGLE + GLUT_RGB)
call fglutCreateWindow(’Fortran GLUT program’)
call fglutDisplayFunc(display)
call fglutMainLoop
end

Fortranski program je sestavljen iz dveh delov: glav-
nega programacrta in podprograma display. Z uka-
zomfglutInit inicializiramo GLUT knjǐznico pod-
programov. Sledi zahteva po vrsti okna. S konstantama
GLUT SINGLE in GLUT RGB povemo, dǎzelimo
okno z eno ravnino tribarvnega RGB prostora. spre-
menljivka window hrani številko okna, ki jo naredi
fglutCreateWindow in hkrati pove OS, kaǩsen naj

bo napis na oknu. Okenskemu sistemu moramoše do-
povedati, katere dogodke bo program prestrezal. Za po-
dani primer je to le prikaz vsebine okna. S klicem pod-
programafglutDisplayFunc prijavimo OS, da naj
pǒsilja sporǒcila za izris, knjǐznici GLUT pa s tem do-
povemo, da ob zahtevi za ponovni izris pokliče podpro-
gramdisplay .

Zadnji klic v glavnem programu je vedno
fglutMainLoop . Ta podprogram se konča le
takrat se koňca celoten program. Koťze ime pod-
programa govori, je to glavna zanka programa, v
kateri GLUT sistematǐcno sprejema sporočila in
kli če nǎse podprograme za dogodke. V podanem
primeru je to le en tip dogodkov, ki je tudi najbolj
uporaben -display . Ostale tipe dogodkov pa
fglutMainLoop obdeluje na privzeti nǎcin, ki je
lahko preprosto ignoriranje sporočila ali pa klicanje
vgrajenega podprograma, ki obdela sporočilo. Izgled
glavnega programa je tako običajno zelo podoben za
vse tipe GLUT programov v katerem si sledijo ukazi v
naslednjem zaporedju:

1. Vklju či definicije konstant GLUT z ukazom
include ’GL/glut.h’

2. Dopovej Fortranu, da so imena, kot je npr.
display , podprogrami in ne spremenljivke. To
se naredi z ukazomexternal

3. Inicializiraj GLUT

4. Nastavi parametre okna (položaj, velikost, tip, bi-
tne ravnine, pomnilnik)

5. Naredi okno in ga poimenuj

6. Prijavi podprograme, ki jih bo program izvajal ob
dogodkih. Obvezno prestrezanje je le zadisplay.
Ostali so poljubni.

7. Nastavi lastnosti OpenGL stroja. To so običajno
ukazi fglEnable ali pa kaǩsna nastavitev lǔci,
materialov in obnǎsanja GL stroja. V tem področju
se obǐcajno nastavi tudi ostale spremenljivke, ki
niso neposredno vezane na OpenGL, ampak na
samo delovanje programa, ki poleg prikaza delaše
kaj drugega.

8. Zadnji je klic fglutMainLoop , iz katerega se
program vrne, ko zapremo okno. Ob tem glavni
program koňca.

2



2.3 Izris v jeziku OpenGL

V podprogramudisplay se morajo torej nahajati
ukazi, ki nekaj narǐsejo v okno. To so ukazi v jeziku
OpenGL ali kraǰse v jeziku GL. Vsi podprogrami ali
funkcije v GL imajo predpono pri imenugl oziroma za
Fortranfgl. Te predpone so potrebne zaradi možnega
prekrivanja z imeni v drugih knjižnicah in jezikih. Za
razumevanje jezika se lahko funkcije razlaga brez teh
predpon, saj je OpenGL koncipiran tako, da so tipi ar-
gumentov za vse jezike podobni. Za posamezen jezik se
predpostavi prefiks (za Fortran je tofgl , za C pagl )

Podprogramdisplay torej skrbi za izris vsebine
okna. Z ukazomClear se brǐse celotno podrǒcje okna.
Kaj konkretno se briše povemo z argumentom. V našem
primeru je toGL COLORBUFFERBIT , kar pomeni
brisanje vseh tǒck v barvnem pomnilniku.

Z ukazomColor se nastavi trenutna barva grafičnih
gradnikov, ki se bodo izrisovali v nadaljnih ukazih.
Kot argument se podaja barva v RGB komponen-
tah. Obǐcajno imajo GL ukazi na koncu imena tudi
oznako tipa argumentov, ki jih je potrebno podati pri
klicu podprograma. To je običaj za skoraj vse jezike.
Tako imamo za Fortran in za C pri podprogramih, kot
končnico imenǎse oznako ǒstevilu argumentov in tip
argumentov. PodprogramfglColor3f torej pomeni, da
zahteva podprogram tri argumente tipafloat , kar je
ekvivalentno v fortranu tipureal ali real*4 . Najbolj
uporabljane GL ukaze lahko torej podamo z različnimi
tipi argumentov za isti ukaz. Izbor tipa argumentov je
odvisen od programerja in njegovih zahtev. Tako so ar-
gumenti za isto funkcijo izbrani glede na priročnost. Za
podani primer imamo ukazVertexv dveh oblikah in isti
tip argumentov.Vertex2f pomeni, da podajamo kor-
dinate vozlǐsča z dvema argumentoma. Tipi argumen-
tov (koňcnic) so naslednji:

f V jeziku C float in real ali real*4 za Fortran

d double za C in real*8 za Fortran

i integer

s short integer v C-ju ali integer*2 za F

Namesto fiksiranegǎstevila argumentov obstajajo tudi
funkcije, ki imajo podan argument v obliki vektorja. Za
to se uporabi koňcnicav . Nekaj primerov koňcnic:

3f Sledijo trije argumenti realniȟstevil

3i Sledijo trije argumenti celiȟstevil

3fv Sledi vektor treh realniȟstevil

Ker je GL v osnovi 3D pomeni, da so argumentiče
sta podani le dve koordinatix in y v ravnini z = 0.
Torej je to ekvivalentno klicu podprogramaVertex(x, y,
0). V nǎsem primeru imamo tudi nastavitev vozlišča z
ukazomVertex3f(0.8,0.8,1.0) , ki podaja vse
tri koordinate v prostoru. Koordinataz = 1 je torej
podana, vendar je zaradi privzetega začetnega ortograf-
skega pogleda v prostor ravnine (x,y) koordinataz po
globini neopazna. Če pa bi bila projekcija perspek-
vivna in ne ortogonalna, bi opazili tudi vpliv koordi-
natez. Raznolikost tipov argumentov se pokaže ravno
pri podajanju vozlǐsč, saj obstajajo naslednjji podpro-
grami: glVertex2d, glVertex2f, glVertex2i, glVertex2s,
glVertex3d, glVertex3f, glVertex3i, glVertex3s, glVer-
tex4d, glVertex4f, glVertex4i, glVertex4s, glVertex2dv,
glVertex2fv, glVertex2iv, glVertex2sv, glVertex3dv, gl-
Ver tex3fv, glVertex3iv, glVertex3sv, glVertex4dv, glVer
tex4fv, glVertex4iv, glVertex4sv. In vse to za en sam
ukaz.

Namen velikegǎstevila istih podprogramov za isto
funkcijo je opustitev pretvarjanja tipov in s tem pisanje
bolj razumljive in hitreǰse kode. V jeziku C++ ali Java,
ki pa sam dodaja ustrezne argumente k imenom funkcij,
pa bi lahko obstajal le en podprogram (npr.glVertex),
jezik pa bi sam dodal ustrezne končnice in s tem klical
ustrezno funkcijo.

Izris grafǐcnih elementov risbe se v GL podaja med
ukazomaglBegin in glEnd. Predmet risanja podamo
kot argument v ukazuBegin . Na splǒsno velja, da
funkcije brez koňcnic zahtevajo en sam argument s kon-
stanto, ki je podana vheaderdatoteki s koňcnico .h
in se vkljǔcuje za zǎcetek programske enote s stavkom
include ’GL/fgl.h’ . Za fortran je programska
enota vsak podprogram, zato moramo pri vsakem pod-
programu, ki uporablja te konstante, na začetku dopisati
še vkljǔcevanje teh konstant. Za C je modul.cdatoteka,
in ni potrebno vkljǔcevanje definicij konstant za vsak
podprogram, kot je to nujno za Fortran.

Vse te GL konstante, ki so napisane vfgl.h in fglu.h
imajo standardno predponoGL in je za vse jezike eno-
znǎcen. Ker pa Fortran ne zahteva deklaracje spremen-
ljivk in ima implicitno definirane tipe je prav mǒzno,
da se zatipkamo pri imenu konstante, kar za Fortran
pomeni realnǒstevilko z vrednostjo 0.0. Da se izo-
gnemo takim tězavam, se priporǒca ukazimplicit none,
s katerim izkljǔcimo predpostavljene tipe in moramo za
vsako spremenljivko povedati, kakšnega tipa je.̌Zal pa
F77 ne omogǒca prototipov tako, da jěse vedno po-

trebna pazljivost, kaǩsne tipe podajamo kot argumente
podprogramom. Posebno to velja za podprograme
GLU, ki običajno nimajo tako razvejanih možnosti ar-
gumentov, kot knjǐznicaGL.

Zadnji ukazglFlush dopove GL stroju naj vse te
ukaze, ki jih je sprejel do sedaj, spravi iz svojih in-
ternih pomnilnikov v okno okenskega sistema. Ker
imamo v nǎsem primeru le enostaven izris, smo se
odločili le za en slikovni pomnilnik (GLUTSINGLE),
ki je primeren le za statične slike. Za aplikacije pri
katerih se vsebina zaslona pogosto spreminja, je pri-
merneje uporabiti okno z dvema grafičnima pomnil-
nikoma GLUTDOUBLE. Prednost slednjega je v tem,
da v en pomnilnik rǐsemo, drugega pa prikazujemo.
Rišemo v ravnino, ki je v ozadju. Ob koncu risa-
nja pa le zamenjamo ravnini. Ker pa je to odvisno
od sistema se ukaz za zamenjavo risalnih ravnin ime-
nujeglutSwapBuffers . Prednost takega načina se
pokǎze pri animacijah.

3 Geometrijski primitivi

Uporabiti je mogǒce le enostavne primitive. To pa pred-
vsem zaradi zahtevane hitrosti in enostavosti izdelave
strojnega pospeševanja. Lǒcimo tri vrste teh enostavnih
primitivov:

• točke ali pike

• črte

• ploskvice konveksnega tipa

Bolj zahtevne predstavitve izvedemo z kombiniranjem
teh primitivov. Tako krivulje razlǐcnih tipov aproksi-
miramo z lomljenkami, povřsine pa s ploskvicami. Za
najbolj ražsirjene kompleksne tipe sěze nahajajo pod-
programi v knjǐznici GLU. Obstajajo tudi mǒznosti se-
stavljenih enostavnih gradnikov začrte in ploskvice. Za
črte poznamo tako naslednje možnosti:

GL LINES Pari vozlǐsč podajajo posamezne se-
gmente

GL LINE STRIP Zaporedje povezanih vozlišč po-
daja lomljenko

GL LINE LOOP Lomljenka se zakljǔci tako, da
pověze prvo in zadnje vozlišče.

Konstante so podane kot argument za podprogramBe-
gin. Za ploskve se podaja več točk. Najenostavnejše

ploskve so trikotniki. Mǒzni soše ravninskǐstirikotniki
in konveksni ravninski mnogokotniki. Enostavne ele-
mente lahko podajamo tudi v pasovih in trikotnike v
pahljǎcah:

GL TRIANGLES Tri vozlišča za en trikotnik

GL TRIANGLE STRIP Pas trikotnikov. Tri vo-
zlišča za prvi in nato vsako nadaljnje vozlišče k
preǰsnjemo trikotniku doda nov trikotnik.

GL TRIANGLE FAN Pahljǎca: vsako dodatno vo-
zlišče naredi dodaten trikotnik v smislu dežnika.

GL QUADS Ravninskištirikotnik se podaja šstirimi
vozlišči.

GL QUAD STRIP Dodatni štirikotniki gradijo pas z
dodajanjem parov vozlišč.

GL POLYGON En sam konveksni mogokotnik po-
ljubnegaštevila vozlǐsč.

Za nesestavljeni tipe gradnikov lahko medBeginin End
podamo tudi věc vozlišč. Tip gradnika se pri pri tem
avtomatsko ponovi. MedBeginin Endse lahko upora-
bljajo še ukazi za barvoglColor in normaleglNormal.
Ti ukazi nastavljajo trenutno stanje, ki velja za vsa na-
slednja vozlǐsča. Primer podprograma za prikaz enega
trikotnika v ravnini je naslednji:

subroutine display
implicit none
include ’GL/fgl.h’
call fglClear(GL_COLOR_BUFFER_BIT)
call fglBegin(GL_TRIANGLES)
call fglColor3f(1.0, 0.0, 0.0)
call fglVertex2f(-1.0, -1.0)
call fglColor3f(0.0, 1.0, 0.0)
call fglVertex2f(0.0, 1.0)
call fglColor3f(0.0, 0.0, 1.0)
call fglVertex2f(1.0, 0.0)
call fglEnd
call fglFlush
end

Pred vsako tǒcko je podanǎse trenutna barva. Izri-
sani trikotnik tako ni enotne barve ampak se njegova
notranjost preliva iz ene skrajne barve v drugo. Re-
zultat prikazuje slika 1. Uporaba različnih barv v vo-
zliščih mogǒce ni posebno uporabna. Za podajanje
normal pa je obǐcajno potrebno, da so normale v vo-
zliščih razlǐcne. Razlǐcne normale v vozliščih nastajajo
povsod tam kjer imajo ploskvice skupen rob, za kate-
regaželimo, da ima gladek prehod. To pa je povsod
tam, kjer aproksimiramo ,,gladko“ površino z osnov-
nimi gradniki. Slika 2 kǎze splǒsno postavitev treh tǒck
v prostoru. Normalo za trikotnik z vozlišči ~r0, ~r1, ~r2

Slika 1: Trikotnik s podanimi barvami v vozliščih

Slika 2: Normala

izračunamo z vektorskim produktom

~n =
(~r1 − ~r0)× (~r2 − ~r0)
|(~r1 − ~r0)× (~r2 − ~r0)|

Imenovalec zgornje enačbe je doľzina vektorja~n.
Normala je pravokotna na razliko vektorjev, ki podajajo
vozlišče gradnika.

4 Geometrijske transformacije

Osnova vseh grafǐcnih knjižnic so tudi osnovne geome-
trijske transformacije, kot so:

Translate(x, y, z) Premik v smeri vektorja

Rotate(fi, x, y, z) Rotacija zafi stopinj okoli osi po-
dane z (x, y, z)

Scale(x, y, z)Skaliranje po posameznih oseh

Ukazi za transformacije se ne smejo pojavljati medBe-
gin/End, saj bi to pomenilo, da se transformacija spre-
minja med izrisom. Geometrijske transformacije nam
pomagajo pri modeliranju, saj lahko podajamo vozlišča
gradnikov v nekem poljubnem koordinatnem sistemu.

To je lahko svetovni koordinatni sistem ali lokalni koor-
dinatni sistem. Za primer izberimo izris krivuljey(x) =
sin(x) v jeziku GL. Kot smože opazili, je prednasta-
vljeno okno v GLUT obliki kvadrata, velikosti (-1,-1)
do (1,1). Vsega skupaj torej dve enoti. V zaslonskih
koordinatah je prednastavljena velikost okna300× 300
pikslov. Za nas je pomembno, da sinus narišemo v me-
jah od -1 do 1. Vzemimo primer, ko predvidimoštevilo
točk. Podprogram za izris je naslednji:

subroutine display
include ’GL/fgl.h’
call fglClear(GL_COLOR_BUFFER_BIT)
call fglBegin(GL_LINE_STRIP)
do i=0,10

y = sin((i-5)/5.0*3.14)
call fglVertex2f((i-5)/5.0, y/3.14)

end do
call fglEnd
call fglFlush
end

Da smo spravili nǎsih 11 tǒck lomljenke v okvir -
1, 1 je bilo potrebno premakniti koordinatni sistem osi
x za 5, ga natǒse skalirati tako, da smo iz območja
[0,10] dobili obmǒcje [-3.14, 3.14]. Čeprav smo za
izračun koordinate y potrebovali na osi x območje [-
3.14, 3.14] pa je potrebna osx za izris v obmǒcju [-
1,1]. Zato pri izrisu podajamo osx tako, da ponovno
porǎcunavamo obmǒcje [0,10] v obmǒcje [-1,1], tako
da i -ju oďstejemo 5 in delimo z 5. Lahko bi tudi delili
s 5 in oďsteli 1. Nekoliko bi poenostavili stvari,̌ce bi
imeli vsaj en kordinatni sistem̌ze takoj uporaben. Re-
cimo osx. Zanka se nekoliko poenostavi,še vedno pa
je potrebno vse koordinate pomanjšati za 3.14 oziroma
poskalirati.

do x=-3.14, 3.14, 0.6
y = sin(x)
call fglVertex2f(x/3.14, y/3.14)
end do

Bolj razumljivo bi bilo risati kar v lokalnem koor-
dinatnem sistemu in prednastaviti pomanjšavo modela.
Za pomanǰsavo uporabimo ukaz za skaliranje, ki posa-
mezne koordinate množi s konstanto 1/3.14, preden se
izriše. Podprogram za izris je naslednji:

subroutine display
include ’GL/fgl.h’
call fglClear(GL_COLOR_BUFFER_BIT)
call fglScalef(1/3.14, 1/3.14, 1.0)
call fglBegin(GL_LINE_STRIP)
do x=-3.14, 3.14, 0.6
y = sin(x)
call fglVertex2f(x, y)
end do
call fglEnd
call fglFlush
end

Prednost takega načina razmǐsljanja se pokǎze,že ko
želimo pod sinusom narisatiše krivuljo kosinusa. Se-
veda ni mǒzno obeh krivulj risati zGL LINE STRIPv

isti zanki. Zato se odlǒcimo za ponovno risanje v lo-
kalnem koordinatnem sistemu in prednastavimo pomik
navzdol za 1.5 enote.

subroutine display
include ’GL/fgl.h’
call fglClear(GL_COLOR_BUFFER_BIT)
call fglScalef(1/3.14, 1/3.14, 1.0)
call fglBegin(GL_LINE_STRIP)
do x=-3.14, 3.14, 0.6
y = sin(x)
call fglVertex2f(x, y)
end do
call fglEnd
call fglTranslatef(0.0, -1.5, 0.0)
call fglBegin(GL_LINE_STRIP)
do x=-3.14, 3.14, 0.6
y = cos(x)
call fglVertex2f(x, y)
end do
call fglEnd
call fglFlush
end

Podani program za kosinus ne nastavlja ponovno ska-
liranja, saj je ukažze pred tem nastavil pomanjšavo.
Translacija za -1.5 se izvede v koordinatnem sistemu
kosinusa. Splǒsen napotek za razumevanje transforma-
cije vsakega vozlišča je, da se za podano koordinato
upǒstevajo transformacije, kot so napisane od spodaj na
vzgor. Transformacija, ki se izvede zadnja je torej na-
pisana na prvem mestu v programu. Tak način transfor-
miranja tǒck nam omogǒca enostavnejše modeliranje.
Koordinatay kosinusa se izrǎcuna tako, da se pred izri-
som najprej vsaki tǒcki y prišteje translacija -1.5 in po-
tem sěse izvede skaliranje tako, da se ta vmesna točka
pomnǒzi še z 1/3.14.

4.1 Nadzor transformacijske matrike

OpenGL pa za izrǎcun koordinat ne hrani vse zgodo-
vine posameznih transformacij za nazaj, saj bi bilo to
računsko potratno. Vse te transformacije, ki jih v po-
ljubnem zaporedju navajamo v programu, popravljajo
transformacijsko matriko. OpenGL ima le dve aktivni
transformacijski matriki, ki jih uporablja za poračun ko-
ordinat. Prva matrika je modelna, druga pa je projekcij-
ska. Mi bomo uporabljali le modelno transformacijo
in upǒstevali, da projekcijska matrika omogoča prikaz
ravnine(x, y) v podrǒcju [-1,1]. Modelna matrika je
tudi stalno aktivna,̌ce se ne izbere projekcijsko.

Modelna matrika se ob vsakem klicu transforma-
cijskega podprograma popravi. Začetna oblika mo-
delne matrike je enotska. Vsak klic podprogramaTran-
slate, Scalein Rotatepa matriko popravi tako, da so
upǒstevane vse prejšnje transformacije in nad njimiše
novo podana transformacija. Matrika je torej stalna,

kar se kǎze tudi v napaki prejšnjega programa za iz-
ris sinusa in kosinusa, ki je pri vsakem ponovnem izrisu
trikrat manǰsi. To lahko preverimo tako, da okno pre-
krijemo s kakim drugim oknom in ga potem ponovno
odkrijemo. Kotže omenjeno, je na začetku programa
matrika enotska. S podprogramomfglLoadIdentityna
zǎcetku bi lahko to tudi zagotovili ob vsakem izrisu.

Za bolj zahtevne transformacije je potrebno matriko
zǎcasno shraniti in obnoviti. OpenGL ima v ta namen
poseben pomnilnik v obliki sklada, v katerega lahko
shranjujemo trenutno transformacijsko matriko. V po-
mnilniku oblike LIFO (Last In, First Out) je prostora
je za najmanj 32 matrik. Pomnilnik si lahko predsta-
vljamo kot hladilnik, v katerega shranjujemo matrike.
Matriko, ki jo želimo shraniti, potisnemo na začetek in
to tako, da vse ostale matrike potisnemo malo naprej
na polici. Ko nekajželimo iz hladilnika, je to lahko
le zadnja matrika.Če želimo predzadnjo, moramo po-
prej vzeti zadnjo. Za shranitev trenutne matrike se upo-
rabi glPushMatrix , za ponastavitev iz sklada pa upo-
rabimoglPopMatrix . Izkǎze se, da je za modeliranje
taka oblika pomnilnika povsem primerna.

Za primer vzemimo primer kocke sestavljene iz
šestih ploskev. Za izris kvadrata obstajaže kraǰsa funk-
cija glRectf(x1, y1, x2, y2) za ravninoz =
0. Če želimo imeti kvadrat v poljubni ravnini, pa upo-
rabimo transformacije.

subroutine kvadrat(i)
real r(6), g(6), b(6)
data r /1,0,0,1,1,1/, g /0,1,0,1,0,0/
data b /0,0,1,0,1,1/
call fglPushMatrix
call fglColor3f(r(i), g(i), b(i))
call fglTranslatef(0.0, 0.0, 1.0)
call fglRectf(-1.0, -1.0, 1.0, 1.0)
call fglPopMatrix
end

subroutine display
implicit none
include ’GL/fgl.h’
call fglClear(GL_COLOR_BUFFER_BIT+GL_DEPTH_BUFFER_BIT)
call fglPushMatrix
call fglRotatef(30.0, 1.0, 0.0, 0.0)
call fglRotatef(30.0, 0.0, 1.0, 0.0)
call fglScalef(0.5, 0.5, 0.5)
call kvadrat(1)
call fglRotatef(90.0, 0.0, 1.0, 0.0)
call kvadrat(2)
call fglRotatef(90.0, 0.0, 1.0, 0.0)
call kvadrat(3)
call fglRotatef(90.0, 0.0, 1.0, 0.0)
call kvadrat(4)
call fglRotatef(90.0, 1.0, 0.0, 0.0)
call kvadrat(5)
call fglRotatef(180.0, 1.0, 0.0, 0.0)
call kvadrat(6)
call fglPopMatrix
call fglFlush
end

program kocka
external display
include ’GL/fglut.h’
include ’GL/fgl.h’
call fglutinit
call fglutInitDisplayMode(GLUT_SINGLE+GLUT_DEPTH)
call fglutCreateWindow(’Fortran GLUT program’)
call fglutDisplayFunc(display)
call fglEnable(GL_DEPTH_TEST)
call fglutmainloop
end

Podprogramkvadrat je narejen tako, da riše tran-
sliran kvadrat v ravniniz = 1. To lahko razumemo
kot nov primitiv, saj par ukazov Push/Pop ne popravlja
transformacije ob klicu podprograma.̌Sest stranic se
riše z rotacijo osnovne stranice okoli osi y in x. Mo-
delna matrika se shani z začetnim ukazomPushin po-
tem ponovno obnovi z ukazomPop.

4.2 Globinski pomilnik

Da se ploskve v prostoru pravilno izrisujejo tudi takrat,
ko rišemo ploskvice za drugimi, je potrebno uporabiti
globinski pomnilnik aliz-buffer. To pa mora omogǒcati
že sam okenski sistem, zato je potrebno tak način
prikaza zahtevatǐze pri fglutInitDisplayMode
in kasneješe dopovedati GL stroju, da poleg barve
točk na zaslonu shranjujěse koordinatoz v svoj
pomnilnik. S tem pomnilnikom GL ob rasteriza-
ciji lika za vsako tǒcko ugotovi, če je že kaǩsna
točka po globini pred njim in jo zato ne riše. Z
ukazomfglEnable(GL DEPTHTEST) se zahteva
izračunavanje globine, ki jo je potrebno tako kot
barvo pred vsakim zǎcetkom risanja pobrisati z ukazom
fglClear .

Slika 3: Kocka brez spodnjega in zgornjega pokrova
(kvadrat(5) in kvadrat(6)

Če rǐsemo zaprte modele, potem notranjosti ni možno
videti. Primer odprtega modela kaže slika 3. V ta-
kih primerih se ob uporabi prostorskega pomnilnika

običajno kar polovica ploskvic modela prekrije v ce-
loti in kasneje na zaslonu ni vidna. Skupna značilnost
vseh teh ploskvic, ki se prekrijejo je, da imajo normalo
povřsine negativno (nz < 0). Da se izogemo nepotrebni
rasterizaciji teh ploskvic, vkljǔcimo GL CULL FACE.
Da pa bo izlǒcanje delovalo, mora imeti GL podatek za
normalo povřsine, ki jo je potrebno podati pred podatki
v vozliščih. Za pravilno delovanje globinskega pomnil-
nika je potrebna tudi nastavitev projekcijske matrike kot
je to opisano v§4.4.

4.3 Animacija

Imejmo primer animacije vozil na avtocesti. Predsta-
vljeno bo cestǐsče v eno smer z dvema pasovoma, vo-
znim in prehitevalnim. Vozila imajo začetni polǒzaj in
hitrost. Opazujemo vozišče doľzine 500 metrov. Hi-
trost vozila med vǒznjo se ne spreminja. Spreminja se
le polǒzaj vozil (x, y) na cestǐsču, ki jih izriše podpro-
gramvozilo .

subroutine display
implicit none
include ’GL/fgl.h’
common /vozila/ y(5), v(5)
real y, v, pas
integer i
data y /0,50,120,170,200/
data v /50,30,45,31,33/
call fglClear(GL_COLOR_BUFFER_BIT)
call fglPushMatrix
call fglRotatef(-45.0, 0.0, 0.0, 1.0)
call fglTranslatef(0.0, -1.0, 0.0)
call fglScalef(0.004, 0.004, 0.004)
call fglColor3f(0.0, 0.0, 0.0)
call fglRectf(-4.0, 0.0, 4.0, 500.0)
call fglTranslatef(0.0, -50.0, 0.0)
do i=1,5

if (i.ne.5 .and. y(i+1)-y(i).lt.10.0) then
pas=-2.0

else
pas = 2.0

end if
call vozilo(y(i), pas)

end do
call fglPopMatrix
call fglutSwapBuffers
end

subroutine vozilo(y, pas)
call fglPushMatrix
call fglColor3f(1.0, 1.0, 1.0)
call fglTranslatef(pas, y, 0.5)
call fglRectf(-2.0, 0.0, 2.0, 6.0)
call fglPopMatrix
end

subroutine ura(n)
common /vozila/ y(5), v(5)
real y, v, dt
dt = 0.1
do i=1,5

y(i)=y(i)+v(i)*dt
end do
call fglutPostRedisplay

call fglutTimerfunc(100, ura, 0)
end

program Mad Max
external display
external ura
include ’GL/fglut.h’
integer window
call fglutInit
call fglutInitDisplayMode(GLUT_RGB+GLUT_DOUBLE)
call fglutCreateWindow(’Avtocesta’)
call fglClearColor(0.0, 0.5, 0.0, 0.0)
call fglutDisplayFunc(display)
call fglutTimerFunc(100, ura, 0)
call fglutMainLoop
end

Pas predstavlja odmik v smerix od sredine cestišča.
Vse enote so v metrih. Vozilo je zaradi sorazmerja nari-
sano nekoliko věcje. Za animacije je primernejša upo-
raba dvojnega pomnilnikaGLUTDOUBLE. S tem se iz-
ognemo tězavam izrisa, saj v trenutku, ko se zgornja
plast izrisuje, nemoteno rišemo v spodnjo plast. Ko je
spodnja plast izdelana z ukazomfglutSwapBuffers, za-
menjamo trenutni prikaz.

Za animacijo, pri kateri je zahtevano točno časovno
zaporedje je primerno uporabiti uro (timer), ki pro-
gram opozori, da je pretekel predpisaničas in da je
potrebno izrǎcunati nov polǒzaj vozil. V nǎsem pri-
meru je podana spremeba vsakih 100 ms in zato nov
položaj v smeriy linearno narǎsča zav(i)dt, kjer je
hitrost podana v metrih na sekundo. Izbor 0.1s za
premik pomeni 1/0.1=10 posnetkov na sekundo, kar
je spodnja meja pri animacijah. Po poračunu novih
položajev pǒsljemo sporǒcilo fglutPostRedisplay, da se
na novo izrǐse scena. Lahko bi tudi neposredno klicali
display , vendar bi bilo potem potrebno zagotovitiše
kompenzacijo hitrosti, saǰze v podprogramu ura izgu-
bimo nekajčasa pri izrěcunu novih polǒzajev. Prostor-
ski pomnilnik v tem primeru ni potreben, saj je zagoto-
vljeno, da se izrisi prekrijejo v pravilnem vrstnem redu.

4.4 Transformacije pogleda

Za zahtevnejše nǎcine gledanja na model je potrebno
nastaviti projekcijo modela iz svetovnih koordinat v
normalizirane oz. zaslonske koordinate. V praksi ob-
stajata dva nǎcina projekcije: ortografska in perspek-
tivna. V tehnǐski predstavitvah se uporablja predvsem
paralelna oz. ortografska projekcija. Le v primeru
animacije, kjerželimo poudariti blǐzino in oddaljenost
določenih objektov, se uporablja tudi perspektivna pro-
jekcija.

OpenGL lǒcuje projekcijsko matriko in modelno ma-
triko zato, da ni potrebno nastavljati projekcije pri vsa-

Slika 4: Zaporedje pretvorbe koordinat vozlišč

kem izrisu. Slika 4 kǎze zaporedje transformacij iz
svetovnih koordinat v zaslonske. Pri risanju modela
običajno zǎcnemo z enotskoModelViewmatriko.

Najpreprosteǰsi nǎcin prikaza, kot je bil prikazan tudi
v dosedanjih primerih je, da stlačimo nǎs model s trans-
formacijami v privzete normalizirane koordinate [-1, 1].
Pri tem nǎcinu sta tako modelna kot projekcijska ma-
trika enotski. Modelna matrika je enotska le na začetku
vsakega risanja, projekcijska pa je konstantna vesčas.
Pri takem nǎcinu ni potrebno preklapljati med trenutno
aktivnima projekcijama. Iňce se zadovoljimo s takim
nǎcinom, potem zadostuje tudi privzeta zaslonska trans-
formacija pri spremembi velikosti okna, ki je pri sis-
temu GLUT le enovrstǐcen ukaz:

call fglViewport (0, 0, width, height)
Nekoliko zahtevnejša je sorazmerna sprememba, ki

ne bo anamorfǐcno popravljala velikosti okna:
subroutine reshape (w, h)
integer w, h
implicit none
include ’GL/fgl.h’
common /viewport/ width, height
integer width, height
real*8 left, right, bottom, top, znear, zfar
width = w
height = h
if (w .ge. h) then

left = -width/(1.0*height)
right = width/(1.0*height)
bottom = -1.0
top = 1.0

else
left = -1.0
right = 1.0
bottom = -height/(1.0*width)
top = height/(1.0*width)

end if
znear = -1.0
zfar = 1.0
call fglViewport (0, 0, width, height)
call fglMatrixMode (GL_PROJECTION)
call fglLoadIdentity
call fglOrtho(left, right, bottom, top, znear, zfar)
call fglMatrixMode(GL_MODELVIEW)
end

Predstavljeni podprogram se priporoča v uporabo za
vse programe, ki pripravljajo model v velikosti [-1, 1]
za ModelView. Če bi želeli dodati modelno transfor-
macijo v reshape, potem za zadnjo vrstico dopišemo

še modelno transformacijo in nato v programu za iz-
ris pred zǎcetkom le obnovimo stanje modelne ma-
trike. Primer animacije 4.3 bi tako imel namesto na-
stavitve modelne transformacije slednje v podprogramu
reshape. Zǎcetna nastavitev modelne matrike pred
zǎcetkom izrisa v podprogramudisplaypa bi bila:

call fglClear(GL_COLOR_BUFFER_BIT)
call fglPushMatrix
... izris
call fglPopMatrix

Takoj za brisanjem zaslona z ukazomPushshranimo
modelno matriko in jo ob koncu ponovno nastavimo na
zǎcetno vrednost. V podprogramureshape, pa modelno
matriko popravljamo:

call fglMatrixMode(GL_MODELVIEW)
call fglLoadIdentity
call fglRotatef(-45.0, 0.0, 0.0, 1.0)
call fglTranslatef(0.0, -1.0, 0.0)
call fglScalef(0.004, 0.004, 0.004)

Tak pristop nekoliko jasneje predstavi program, saj
so vse enote, s katerimi manipuliramo, v programu za
izris v svetovnih oz. modelnih koordinatah. V spošnem
se priporǒca nastavitev projekcije za vse modele, ki
uporabljajo izris ploskev. To pa zaradi tega, ker je pri-
vzeta projekcijska matrika enotska. Poglejmo to na pri-
meru paralelne projekcijeglOrtho(l,r,b,,n,f):

PM =


2

r−l 0 0 r+l
l−r

0 2
t−b 0 t+b

t−b

0 0 2
f−n

f+n
f−n

0 0 0 1

 .

Za primer normalizacijskega prostora v obsegu [-1,1] je
tako matrika paralelne projekcije

PM(−1, 1,−1, 1,−1, 1) =


1 0 0 0
0 1 0 0
0 0 −1 0
0 0 0 1

 ,

kar se razlikuje od enotske prav v koordinatiz. Če
bi projekcijsko matriko ohranili enotsko, potem bi to
pomenilo, da objekt gledamo kot zrcalno sliko zadnje
strani. Nastavitev projekcijske matrike je torej obve-
zna za vse izrise ploskvic po globini, kot tudi za mo-
dele z oseňcenjem. Zaradi tega je tudi program za izris
kocke nelogǐcno postavil v ospredje modro stranico in
ne rděco.

5 Osvetlitev

Do sedaj predstavljeni primeri so uporabljali le sin-
tetične barve. To pomeni, da se barva vsake ploskvice

ne spreminja v odvisnosti od položaja v prostoru. Tak
nǎcin prikaza je uporaben le za omejen nabor prostor-
skih modelov. Neprimeren jěze za vse modele, ki imajo
povřsine sestavljene iz primitivov in te površine niso
ravninske. Za primer kocke (slika 3) je bilo potrebno
za vsako stranico nastaviti svojo barvo, da smo lahko
dobili vtis prostora.Če bi kocko risali le z eno barvo,
potem bi dobili na zaslon le obris.

Za bolj realistǐcen izris je potrebno vkljǔciti
računanje osvetlitve.̌Zal osvetlitev zajema veliko para-
metrov, ki jih je potrebno nastaviti preden lahko karkoli
dobimo na zaslonu. Tako je potrebno nastavljati položaj
in lastnosti lǔci, osvetlitveni model in lastnosti površin
modelov. Za vsako lǔc se lahko tako nastavi 10 lastnosti
in vsaka povřsina ima 5 lastnosti materiala.

Kot predpogoj za pravilno osvetljen model pa je po-
dana normala v vsakem vozlišču vsake ploskvice. Naj-
preprosteǰsi nǎcin pri uporabi osvetlitve je, da parame-
tre luči ne nastavljamo in da uporabimo nastavljanje la-
stnosti materiala površine le z ukazom za barvo. S tem
vpeljemo veliko predpostavk, ki pa so zašolsko rabo
povsem uporabne. Predpostavljena je le ena luč bele
svetlobe s polǒzajem(0, 0, 1) in difuzni odboj svetlobe
na povřsini. Barvo povřsine podajamo kar z običajnim
ukazom za barvo. Program za izris osenčenega modela
kocke je tako v minimalni obliki naslednji:

subroutine kvadrat()
call fglPushMatrix
call fglTranslatef(0.0, 0.0, 1.0)
call fglNormal3f(0.0, 0.0, 1.0)
call fglRectf(-1.0, -1.0, 1.0, 1.0)
call fglPopMatrix
end

subroutine display
implicit none
include ’GL/fgl.h’
call fglClear(GL_COLOR_BUFFER_BIT+GL_DEPTH_BUFFER_BIT)
call fglColor3f(0.7, 0.6, 0.2)
call fglPushMatrix
call fglScalef(0.5, 0.5, 0.5)
call kvadrat
call fglRotatef(90.0, 0.0, 1.0, 0.0)
call kvadrat
call fglRotatef(90.0, 0.0, 1.0, 0.0)
call kvadrat
call fglRotatef(90.0, 0.0, 1.0, 0.0)
call kvadrat
call fglRotatef(90.0, 1.0, 0.0, 0.0)
call kvadrat
call fglRotatef(180.0, 1.0, 0.0, 0.0)
call kvadrat
call fglPopMatrix
call fglFlush
end

program kocka
external display
external reshape
include ’GL/fglut.h’
include ’GL/fgl.h’

call fglutinit
call fglutinitdisplaymode(GLUT_SINGLE+GLUT_DEPTH)
call fglutcreatewindow(’Osencena kocka’)
call fglutDisplayFunc(display)
call fglutReshapeFunc(reshape)
call fglClearColor(1.0, 1.0, 1.0, 1.0)
call fglEnable(GL_LIGHTING)
call fglEnable(GL_LIGHT0)
call fglEnable(GL_DEPTH_TEST)
call fglEnable(GL_COLOR_MATERIAL)
call fglutMainLoop
end

subroutine reshape (w, h)
integer w, h
implicit none
include ’GL/fgl.h’
real*8 l
l = 1.0
call fglViewport (0, 0, w, h)
call fglMatrixMode (GL_PROJECTION)
call fglLoadIdentity
call fglOrtho(-l, l, -l, l, -l, l)
call fglMatrixMode(GL_MODELVIEW)
call fglLoadIdentity
call fglRotatef(30.0, 1.0, 0.0, 0.0)
call fglRotatef(30.0, 0.0, 1.0, 0.0)
end

Ražsirjeni primitiv smo poenostavili tako, da ne vse-
buje věc definicije barve, ampak le geometrijo. Obve-
zno je bilo potrebno podati izračun normale. Za nǎs
primitiv kvadrata je to(0, 0, 1). Program za izris v bi-
stvu ni spremenjen, le da je sedaj transformacija modela
preseljena v podprogram za nastavitev velikosti okna
reshape . V glavnem programu pa je potrebno najprej
vključiti računanje osvetlitveGL LIGHTING, prižgati
je potrebno lǔc št 0, ki ima zǎcetni polǒzaj (0, 0, 1).

Slika 5: Oseňcen model kocke

Z vključitvijo GL COLORMATERIALpa poenosta-
vimo podajanje barve za material površine tako, da vsi

klici podprogramovColor nastavljajo privzeto difuzno
in ambientno barvo površine. Slika 5 prikazuje rezultat
upodabljanja z osvetlitvijo.

6 Tekst

OpenGL sam ne podpira teksta in je zato potrebno upo-
rabiti razne prijeme za izris teksta v prostoru. Možnih
je věc nǎcinov za risanje besedila:

stroke Črke so izrisane šcrtami v prostoru modela

bitmap Črke so izrisane na zaslon

teksture Črke so izrisane rastrsko v prostoru modela

V šolskih primerih so najbolj uporabniže izdelani fonti
v knjižnici GLUT. Možne so naslednjěstevilke fontov:

1. GLUT STROKEROMAN

2. GLUT STROKEMONO ROMAN

3. GLUT BITMAP 9 BY 15

4. GLUT BITMAP 8 BY 13

5. GLUT BITMAP TIMES ROMAN 10

6. GLUT BITMAP TIMES ROMAN 24

7. GLUT BITMAP HELVETICA 10

8. GLUT BITMAP HELVETICA 12

9. GLUT BITMAP HELVETICA 18

Za primer ražsirimo program za izris osenčene kocke
z besedilom na vsaki stranici. Podprogramkvadratkot
argument vzame besedilo. Začetek izpisa premakne
za malenkost višje in zǎcne v koordinatix = −0.8.
Ker pa neželimo, da se besedilo senči je tu potrebno
izklapljanje seňcenja takrat, ko izrisujemo posamezne
črke. Ker sǒcrke v vnaprej dolǒceni velikost, jih je po-
trebno ustrezno pomanjšati s skaliranjem. Podprogram
fglutStrokeCharacter po vsaki izrisanǐcrti sam
nastavi pomik v smeri x zǎsirino izrisaněcrke.

subroutine kvadrat(s)
include ’GL/fgl.h’
character s*(*), c
call fglPushMatrix
call fglTranslatef(0.0, 0.0, 1.0)
call fglNormal3f(0.0, 0.0, 1.0)
call fglRectf(-1.0, -1.0, 1.0, 1.0)
call fglTranslatef(-0.8, 0.0, 0.01)
call fglDisable(GL_LIGHTING)
call fglScalef(0.003, 0.003, 0.003)
call fglColor3f(1.0, 0.0, 0.0)
lenc = len(s)
do i=1,lenc

c = s(i:i)
call fglutStrokeCharacter(1, ichar(c))

end do

call fglEnable(GL_LIGHTING)
call fglPopMatrix
end

subroutine display
implicit none
include ’GL/fgl.h’
real mat(4)
data mat /0.9, 0.6, 0.3, 1.0/
call fglClear(GL_COLOR_BUFFER_BIT)
call fglClear(GL_DEPTH_BUFFER_BIT)
call fglPushMatrix
call fglRotatef(30.0, 1.0, 0.0, 0.0)
call fglRotatef(30.0, 0.0, 1.0, 0.0)
call fglScalef(0.5, 0.5, 0.5)
call fglMaterialfv(GL_FRONT, GL_DIFFUSE, mat)
call kvadrat(’Spredaj’)
call fglRotatef(90.0, 0.0, 1.0, 0.0)
call kvadrat(’Desno’)
call fglRotatef(90.0, 0.0, 1.0, 0.0)
call kvadrat(’Zadaj’)
call fglRotatef(90.0, 0.0, 1.0, 0.0)
call kvadrat(’Levo’)
call fglRotatef(90.0, 1.0, 0.0, 0.0)
call kvadrat(’Spodaj’)
call fglRotatef(180.0, 1.0, 0.0, 0.0)
call kvadrat(’Zgoraj’)
call fglPopMatrix
call fglFlush
end

subroutine reshape(w, h)
include ’GL/fgl.h’
integer w, h
real*8 l
l = 1
call fglViewPort(0, 0, w, h)
call fglMatrixMode(GL_PROJECTION)
call fglLoadIdentity
call fglOrtho(-l,l,-l,l,-l,l)
call fglMatrixMode(GL_MODELVIEW)
call fglLoadIdentity
end

program crta
external display
external reshape
include ’GL/fglut.h’
include ’GL/fgl.h’
call fglutinit
call fglutinitdisplaymode(GLUT_SINGLE+GLUT_DEPTH)
call fglutcreatewindow(’Fortran GLUT program’)
call fglutDisplayFunc(display)
call fglutReshapeFunc(reshape)
call fglEnable(GL_DEPTH_TEST)
call fglEnable(GL_LIGHTING)
call fglEnable(GL_LIGHT0)
call fglClearColor(1.0, 1.0, 1.0, 1.0)
call fglutmainloop
end

Podajanje barve za površino je spremenjeno tako, da
se ne uporabi funkcijeColor ampak normalno funk-
cijo za podajanje lastnosti materialafglMaterialfv .
Rezultat kǎze slika 6.Če bi napisali komentar pred iz-
risomštirikotnika, potem bi bilo vidno besedilo tudi za
ostale (skrite) strani.

Včasih pa rajězelimo, da se besedilo na zaslonu ne
izrisuje rotirano in seňceno, temvěc da se le pojavi na
določenem polǒzaju v prostoru in potem izriše v zaslon-

Slika 6: Oseňcen model kocke z napisi

skih koordinatah. V ta namen uporabimobitmapfonte
in naslednji podprogram za izpis besedila:

subroutine output(x,y,z,s)
character s*(*)
call fglRasterPos3f(x,y,z)
lenc = len(s)
do i=1,lenc

call fglutBitmapCharacter(6, ichar(s(i:i)))
end do
end

Primer izrisa z bitmap fonti kǎze slika 7

Slika 7: Oseňcen model kocke zbitmapnapisi

7 Uporabniški vmesnik

V nadaljevanju so prikazani primeri programov, ki iz-
koriščajo dodatne funkcionalnost knjižnice GLUT za
vnos dodatnih podatkov v program. To je predvsem
uporaba tipk in mǐske.

7.1 Rotacija s tipkami

Rotiramože vgrajeni geometrijski modelčajnika s tip-
kami x, y, z. Vsak pritisk na tipko pověca kot rotacije
za pet stopinj. Podatke o trenutni rotaciji prenašamo
s poljemcommon. Ker izrisujemožični model, pod-
program zareshapeni potreben. Podprogramkeyboard
ob pritisku na tipko dobi tudi informacijo o zaslonskem
položaju mǐske.

subroutine display
implicit none
include ’GL/fgl.h’
common /rotation/ rx, ry, rz
real rx, ry, rz
call fglClear(GL_COLOR_BUFFER_BIT)
call fglColor3f(0.5, 0.4, 1.0)
call fglPushMatrix
call fglRotatef(rx, 1.0, 0.0, 0.0)
call fglRotatef(ry, 0.0, 1.0, 0.0)
call fglRotatef(rz, 0.0, 0.0, 1.0)
call fglutWireTeapot(dble(r))
call fglPopMatrix
call fglutSwapBuffers
end

subroutine keyboard(key,x,y)
common /rotation/ rx, ry, rz
integer key,x,y
print *, ’Key ’, char(key), key, ’ at’, x, y
if (key .eq. ichar(’x’)) rx = rx + 5.0
if (key .eq. ichar(’y’)) ry = ry + 5.0
if (key .eq. ichar(’z’)) rz = rz + 5.0
call fglutpostredisplay
end

program teapot
external display
external keyboard
include ’GL/fglut.h’
integer window
call fglutInit
call fglutInitDisplayMode(ior(GLUT_DOUBLE,GLUT_RGB))
window = fglutCreateWindow(’Use keys x, y, and z’)
call fglutDisplayFunc(display)
call fglutKeyboardFunc(keyboard)
call fglutMainLoop
end

7.2 Miška in inverzna projekcija

Za vsak pritisk gumba miške lahko dobimo poleg ko-
ordinate tudiše stanje gumbov. Naslednji primer pri-
kazuje risanjěcrte v ravnini (x,y) s tem, da je potrebno
zaslonske koordinate pretvoriti nazaj v modelne.

subroutine redraw

implicit none
include ’GL/fgl.h’
common /vertices/ n, vertex(2, 100)
integer n, i
real vertex
call fglClear(GL_COLOR_BUFFER_BIT)
call fglbegin(GL_LINE_STRIP)
do i = 1,n

call fglVertex2f(vertex(1, i), vertex(2, i))
end do
call fglend
call fglFlush
end

subroutine mouse (button, state, x, y)
implicit none
include ’GL/fglut.h’
include ’GL/fgl.h’
include ’GL/fglu.h’
common /vertices/ n, vertex(2, 100)
integer n, i
real vertex
integer button, state, x, y
integer viewport(4)
real*8 mvmatrix(16), projmatrix(16)
real*8 wx, wy, wz ! returned world x, y, z coords
real*8 px, py, pz ! picked window coortinates
integer status

if (button .eq. GLUT_LEFT_BUTTON) then
if (state .eq. GLUT_DOWN) then

call fglGetIntegerv (GL_VIEWPORT, viewport)
call fglGetDoublev (GL_MODELVIEW_MATRIX, mvmatrix)
call fglGetDoublev (GL_PROJECTION_MATRIX, projmatrix)
note viewport(4) is height of window in pixels
px = x
py = viewport(4) - y - 1
pz = 0.0
print *, ’ Coordinates at cursor are ’, px, py
status = fgluUnProject (px, py, pz, mvmatrix,

projmatrix, viewport, wx, wy, wz)
print *, ’World coords at z=0.0 are ’, wx, wy, wz
n = n + 1
vertex(1, n) = wx
vertex(2, n) = wy
call fglutPostRedisplay

end if
end if
end

program main
external redraw
external mouse
include ’GL/fglut.h’
call fglutinit
call fglutinitdisplaymode(ior(GLUT_SINGLE,GLUT_RGB))
call fglutInitWindowSize (500, 500)
call fglutInitWindowPosition (100, 100)
window = fglutcreatewindow(’Click in window’)
call fglutdisplayfunc(redraw)
call fglutMouseFunc(mouse)
call fglutmainloop
end

7.3 Kvaternionska rotacija

Naslednji program prikazuje vrtenje osenčenega
čajnika z mǐsko. V ta namem se uporabiže izdelam
podprogram v jeziku C, ki ga kličemo iz fortrana in
nam omogǒca kvaternionsko rotacijo. Za vrtenje enot-

ske krogle je potrebno zaznati tako začetni pritisk na
gumb (podprogrammouse) kot vse naslednje pomike
miške (podprogrammotion).

subroutine display
implicit none
include ’GL/fgl.h’
common /quaternion/ last(4), cur(4)
real last, cur, m(4,4)
call build_rotmatrix(m, cur)
call fglLoadIdentity
call fglMultMatrixf(m)
call fglclear(GL_COLOR_BUFFER_BIT+GL_DEPTH_BUFFER_BIT)
call fglutSolidTeapot(dble(0.5))
call fglutSwapBuffers
end

subroutine motion (x, y)
include ’GL/fglut.h’
include ’GL/fgl.h’
implicit none
integer x, y
common /quaternion/ last(4), cur(4)
common /mousestart/ beginx, beginy
common /viewport/ width, height
integer width, height
integer beginx, beginy
real last, cur
real p1x, p1y, p2x, p2y
p1x = (2.0*beginx - width)/width
p1y = (height - 2.0*beginy)/height
p2x = (2.0 * x - width) / width
p2y = (height - 2.0 * y) / height
call trackball(last,p1x, p1y, p2x, p2y)
call add_quats(last, cur, cur)
beginx = x
beginy = y
call fglutPostRedisplay
end

subroutine mouse (button, state, x, y)
implicit none
integer button, state, x, y
include ’GL/fglut.h’
include ’GL/fgl.h’
include ’GL/fglu.h’
common /mousestart/ beginx, beginy
integer beginx, beginy
beginx = x
beginy = y
end

subroutine reshape(w, h)
include ’GL/fgl.h’
integer w, h
real*8 l
common /viewport/ width, height
integer width, height
width=w
height=h
l = 1
call fglViewPort(0, 0, w, h)
call fglMatrixMode(GL_PROJECTION)
call fglLoadIdentity
call fglOrtho(-l,l,-l,l,-l,l)
call fglMatrixMode(GL_MODELVIEW)
call fglLoadIdentity
end

program trackballdemo
implicit none
include ’GL/fglut.h’
include ’GL/fgl.h’

include ’GL/fglu.h’
external display
external motion
external mouse
external reshape
integer window
common /quaternion/ last(4), cur(4)
real last, cur
call trackball(cur, 0.0, 0.0, 0.0, 0.0)
call fglutinit
call fglutinitdisplaymode(GLUT_DOUBLE+GLUT_RGB+GLUT_DEPTH)
window = fglutcreatewindow(’Use mouse to rotate’)
call fglutdisplayfunc(display)
call fglutmousefunc(mouse)
call fglutmotionfunc(motion)
call fglutreshapefunc(reshape)
call fglEnable(GL_LIGHTING)
call fglEnable(GL_LIGHT0)
call fglEnable(GL_DEPTH_TEST)
call fglutmainloop
end

Predstavljeni program je sestavljen iz dveh delov.
Kodo za rotacijo v jeziku Ctrackball.c uporabimo
kod zunanje podprograme. Rezultat osenčenega model,
ki je bil obrnjen z mǐsko, prikazuje slika 8.

Slika 8: Oseňcen modeľcajnika

8 Razvojno okolje

Za šolske probleme smo pripravili razvojno okolje, ki
omogǒca prevajanje kode v jezikih F77, C++ in C za
okenski sistem Windows. Razvojno okolje deluje v
nǎcunu ukazne vrstice in nima priloženega integrira-
nega vmesnika. Vsi ukazi za popravljanje programov in
prevajanje se tako podajajo v ukazni vrstici DOS okna
(Start-Programs-Command Prompt).

Osnova okolja je Borlandov C++ prevajalnik, ki ga
lahko dobimo zastonj. Besedilo dogovora uporabe se
nahaja v datotekilinence.txt. Prevajanje v jeziku For-
tran pa dosězemo z pretvorbo fortranske kode v C, nato
sledi prevajanje v C-ju in povezovanje v končni pro-
gram (.exe). Končni program lahko zǎzenemo z DOS

okna ali z dvoklikom na izvřsni program. Poleg No-
vega grafǐcnega okna vsak GLUT program uporabljaše
konzolo za morebiten vnos ali izpis z ukazomaprint
*, ali read *,

8.1 Namestitev

Namestitev je mǒzna z CD-ROMA ali datoteke
bcc-fgl-full.zip . V slednjem primeru je po-
trebno paketno datoteko odpakirati v začasen imenik,
nakar sledi namestitev tako kot iz CD-ROM-a.

1. Za namestitev je na disku C potrebnih 60 MB pro-
stora!

2. Dvolkikni na install.bat

8.2 Dokumentacija

Po namestitvi se navodila z nahajajo v imeniku
c:\bcc55\doc . Priližena so naslednja navodila v
obliki PDF:

redbook-*.pdf OpenGL Programming Guide

opengl-intro.pdf Ta dokument

fgl.pdf OpenGL reference

fglu.pdf OpenGL Utility reference

fglut.pdf GLUT

f2c.pdf Prevajalnik za Fortran

V datoteki se nahaja ta dokument. Iskanje po doku-
mentaciji za OpenGL (fgl.pdf, fglu.pdf ), iz-
vedemo tako, da natipkamo npr.fglVertex3f(

8.3 Prevajanje

Primeri so v imenikuc:\bcc55\examples .
Pred prevajanjem je potrebno odpreti okno DOS

Start-Run-command-OKin nastaviti pot do prevajalni-
kov z ukazom

PATH=\BCC55\bin;%PATH%

Pomik v imenik naredimo z ukazom

c:
cd \bcc55\examples

Pot lahko nastavimo tudi za celoten sistem:Start-
Settings-Control Panel-System-evironment-¿PATHin
dopǐsemoc:\bcc55\bin; na zǎcetku ali koncu ob-
stojěce poti.

Za prevajanje fortranskih datotek ne smemo upora-
biti končnice .f Primer prevajanja zǎcetnega primera
line.f za izrisčrte:

f77 line

Za najzahtevnejši primer čajnika uporabimo hkratno
prevajanje fortranskega in C programa, ki oba med se-
boj tudi pověze v programtblight.exe

f77 tblight trackball.c

Če imamo věc modulov, potem lahkǒze prevedene pod-
programe.obj le povězemo v izvřsno kodo. Primer:

f77 tblight trackball.obj

Prevajanje in povezovanje v jeziku C se izvede z kli-
cem prevajalnikabcc32. Primer:

bcc32 teapot.c

8.4 Urejanje fortranskih in C programov:

Na urejanje kode lahko uporabimo DOS-ov urejevalnik
EDIT ali Windows notepad. Oba imata svoje slabosti;
EDIT ima tězave z dalǰsimi imeni datotek, NOTEPAD
pa nima prikaza trenutne vrstice in ob prvem shranje-
vanju datoteke lepi koňcnico .txt , tako da moramo
datoteko kasneje preimenovati v.f . Kljub slabostim
sta oba urejevalnika primerna zašolske probleme.

V trenutnem imeniku DOS okna odtipkamo izbrani
ukaz:

notepad teapot.f
edit teapot.f

Besedila nalog so objavljena pod novicami in navo-
dili za vaje nahttp://www.lecad.uni-lj.si

